/*
*************************************************************************************
*************************************************************************************
                                              ARDUINO MEGA 2560 MICROCONTROLLER
*************************************************************************************
*************************************************************************************

[bookmark: _GoBack]File: ArduinoWindspeed_ino_Word 
Date: 9/30/15

Written in C for Arduino Mega 2560 board

Notes:

1. Please enter your Time Zone in the Setup Section below.

2. Daylight Saving Time correction is only included for years 2015,2016,2017. For Years >=2018
   please amend the code in the Conversion routine. 

3. Print hard copy in Landscape mode (i.e File, Page Setup, Landscape)

4. This program was written in 2015 by David Goodsell

*************************************************************************************
                                                            INTRODUCTION
*************************************************************************************

 Purpose: This Arduino Mega 2560 is part of a Windspeed measuring project. It translates windspeed 
          into several files which are serially downloaded to a Raspberry Pi for plotting on a 
          small LCD display. It also interfaces with a GPS receiver and converts UTC time to local.
          
 The Mega program flow is as follows:
 
 1. At the start of each GPS second, the SX28 GPS Board sends a burst of time and date data to the  
    Mega for processing.
 2. Also, all day, the Mega measures Windspeed(at 2.250 second intervals) from a Davis rotary cup 
    anemometer mounted on the roof and stores the values in a small array. The small array gets 
    reset at one minute intervals. 
 3. At one minute intervals:
    a. Receive the latest GPS time and date from the SX28 board @9600 baud.
    b. Check the small array (usually there are 25 readings in a minute) for the Maximum Windspeed
       during the past minute.
    c. Store the latest Maximum value in the proper time-address in the large array. This array  
       holds all 1440 readings for the day.    
    d. Send the whole array to the Raspberry Pi for plotting @115200 baud.
    e. Also send a plotting configuration file to the Raspberry Pi which contains various parameters 
       to display on the plotting LCD.
    f. In the Raspberry Pi, the previous plot is replaced by the new plot, i.e. 1440 times a day.
    g. If there is an increase in the Daily Maximum, send an audio alert flag to the Raspberry Pi.
 4. Reset all values at midnight.
 
*/

//***********************************************************************************

  #include <stdio.h>
  #include <string.h>
  #include <LiquidCrystal.h>

 //Global variables and Arrays
  byte maxToday;
  byte oneMinuteMax;
  byte nowPeakWindspeed;
  byte nowAvgWindspeed;
  byte Windspeed;
  byte minuteMax;
  byte numberOfReadings;
  byte oneMinutePulse;
  byte audioAlertFlag;
  byte daylightSavFlag;
  byte azFlag;
  byte year;
  byte month;
  byte minute;
  byte second;
  int day;
  int hour;
  int i;
  int x;
  int TZ;
  int Zmonth;
  int UTCZerohour;
  int SThour;
  int maxTodayOffset;
  int nowPlotPoint;
  int startCounter;
  int Xoffset;
  int Yoffset;
  int oneReadingIndex;
  String time1;
  String time2;
  String time3;
  LiquidCrystal lcd(8,9,4,5,6,7);
  
  char GPSArray[100] = {};        //holds time,date,satellites from SX28 
  byte todayArray[1500] = {};     //holds 1440 readings per day
  byte nowMinuteArray[30] = {};   //holds windspeed readings from past full minute
  
  
//***********************************************************************************
//                                                                            SETUP
//***********************************************************************************

 void setup()
{
  //Please enter your Time Zone(TZ) below:
  //***************************************
  //The conversion of Time Zones other than those listed below has not been confirmed.
  //Please adapt the code to fit your International Time Zone.
  
  // -4 = AST  (Atlantic)
  // -5 = EST  (Eastern)
  // -6 = CST  (Central)
  // -7 = MST  (Mountain, plus Arizona w/Daylight Saving)
  // -8 = PST  (Pacific)
  // -9 = Alaska
  // -10 = Hawaii  (w/o Daylight Saving Time)
  // -77 = Arizona (w/o Daylight Saving Time)
  
  TZ = -8;
  
  ClearTodayArray();
  TCCR5A = 0;               //timer5 off
     
  //2x16 Blue LCD Setup
  lcd.begin(16,2);          //start the LCD library
  lcd.setCursor(0,0);
  lcd.print("NowWS:");
  lcd.setCursor(0,1);
  lcd.print("NowMx:");
  
  delay(45000);             //allow 45 secs for Raspi to bootup
  
  attachInterrupt(0,ProcessoneMinutePulse, FALLING);   //turn on interrupt0
  
}

//***********************************************************************************
//                                                       THE BIG LOOP, 24/7
//***********************************************************************************

void loop()
 { 
   if (oneMinutePulse == 1)            //test for Interrupt0(INT0), Pin 2
      {
       ReadAndConvertUTCTime();       //read time/date in Serial2 buffer, from SX28, then convert
       ClearArrayAtMidnight();        //to local time depending on Time Zone
       InitializeMaxValues();          //at bootup, initialize time and date for maxToday
       numberOfReadings = 25;         //occasionally 26 readings are made which inflates the MinMax  
       UpdateDailyMax();
      //***********************
       SendConfigAndDataFiles();      //send data and config files to RPi 
      //*********************** 
       ResetMinuteArrayData();
       oneMinutePulse = 0;            //reset interrupt flag
       oneReadingIndex = 0;           //reset index for MinuteArray
       numberOfReadings = 0;
       oneMinuteMax = 0;
       ResetNowMaxOnLCD();
       Serial1.flush();               //flush the serial buffer
       attachInterrupt(0,ProcessoneMinutePulse, FALLING);    //start interrupt0 again
       }
   else
       {
       //if no Minute interrupt, continue collecting new windspeed values at 2.250 sec intervals
       MeasureWindspeed();                //this is the big one
       UpdateDailyMax();
       PrintNowWSToLCD();                 //print very latest 2.250 second Windspeed to small LCD
       }                                  //for casual viewing
   
 }
   
 //***********************************************************************************
 //                                                                      FUNCTIONS
 //***********************************************************************************
 
 //FUNCTION: SERVICE INTERRUPT 0 
 //******************************
  void ProcessoneMinutePulse()
 //The problem is that the interrupt comes in right away and is noted, but the program
 //goes back to the delay(2250) line and has to finish it before anything else happens.
 //In the meantime, the time/date burst from the SX28 comes in and is lost.
 //Therefore, the trick is to open the Serial1 port USART right away and let it capture
 //the burst. Then it can process the data when the current delay(2250) is finished. 
   {
    noInterrupts();       //turn interrupt off 
    Serial1.flush();      //flush the serial buffer
    Serial1.begin(9600);
    oneMinutePulse = 1;
   }
   
 
 //FUNCTION: INITIALIZE MAX TODAY TIME AND VALUE AT BOOTUP
 //********************************************************
 void InitializeMaxValues() 
   { 
    static byte startCounter; 
    if (startCounter == 0)           
       {
       startCounter++;
       maxToday = 0;
       oneMinuteMax = 0;
       ResetMinuteArrayData();
       MaxWindspeedTime();        //calculates current MinuteTime, returns time2 
       }  
   }
   
     
 //FUNCTION: RESET NowMax ON LCD
 //*****************************
 void ResetNowMaxOnLCD() 
 {
  lcd.setCursor(6,1);     //lcd(position#,line#)
  lcd.print("00");        //reset NowMax value, for past minute
 }
  
       
 //FUNCTION: PRINT NowWS TO LCD
 //****************************
 void PrintNowWSToLCD() 
 {
   lcd.setCursor(6,0);                           
     if (Windspeed <= 9)
       {
       lcd.print("0");
       }
   lcd.print(Windspeed);
 } 
 
 
 //FUNCTION: CLEAR todayArray
 //**************************
 void ClearTodayArray()
 {
  for (x=0;x<=1499;x++) 
    {
    todayArray[x] = 0;
    } 
 } 
 

//FUNCTION: CLEAR todayArray at Midnight
//***************************************
void ClearArrayAtMidnight()
{
  if ((hour == 0) and (minute == 0))
  {
    ClearTodayArray();
    maxToday = 0;
    maxTodayOffset = 1;
    time2 = "12:00am";
  }
}

 
 // BIG FUNCTION: SEND CONFIG AND DATA FILES TO RASPBERRY PI
 //**********************************************************
 void SendConfigAndDataFiles()
 {
  Serial2.begin(115200);
  
 //All numbers are two digits plus a semi-colon delimiter 
     
 //Send time and date
 //******************
  if (month <= 9)
    {
    Serial2.print("0");
    }
  Serial2.print(month); Serial2.print(";");
  
  if (day <= 9)
    {
    Serial2.print("0");
    }
  Serial2.print(day); Serial2.print(";");
    
  Serial2.print(year); Serial2.print(";");
  
  if (hour <=9)
    {
    Serial2.print("0");
    }
  Serial2.print(hour); Serial2.print(";");
  
  if (minute <=9)
    {
    Serial2.print("0");
    }
  Serial2.print(minute); Serial2.print(";");
  
  if (nowPeakWindspeed <=9)
    {
    Serial2.print("0");
    }
  Serial2.print(nowPeakWindspeed); Serial2.print(";");
  
  if (maxToday <=9)
    {
    Serial2.print("0");
    }
  Serial2.print(maxToday); Serial2.print(";");
  
  
  //Send nowPlotPoint (the x-axis number of the latest windspeed measured, 0000 to 1440)
  //***********************************************************************************
 
 //Sending as two groups
  // 1. hundreds (00-14) + ";"
  // 2. tens & units (00-99) + ";"
  
  //Send flag to Raspi to indicate increase in Max Windspeed for the day, which triggers audio tone
  if (audioAlertFlag == 1)    //send to Raspi, 99 = no alert, 98 = alert
    {
    Serial2.print("98;");     //the "8" is in the Raspi at data[22]
    audioAlertFlag = 0;
    }
  else
    {  
    //Serial2.print("98;");    //test only, send audioAlertFlag each minute
    Serial2.print("99;");    //leading characters, "PlotPoint"
    }
  
  //Send LatestPlotPoint to Raspi gnuplot stops plotting at the latest point
  //*************************************************************************
  int k = nowPlotPoint;   //0000 to 1440 Example: 09;14; = Raspi: data[24],[25],[26],[27],[28],[29]
  
  // k =0 to 9
  //**********
  int z;
  if (k <=9)
  {
    Serial2.print("00;"); Serial2.print("0"); Serial2.print(k); Serial2.print(";");
  }
  //k = 10 to 99
  //************
 else if (k <=99)
  {
    Serial2.print("00;"); Serial2.print(k); Serial2.print(";");
  }
  //k = 100 to 999
  //**************
  else if (k <=999)
    {
    z = int(k/100);   //getting the hundreds
      if (z <=9)
        {
        Serial2.print("0"); Serial2.print(z); Serial2.print(";");
        }
      else
        {
        Serial2.print(z); Serial2.print(";");
        }
     
    z = (k - (z*100));  //getting the tens and units
        if (z <=9)
        {
        Serial2.print("0"); Serial2.print(z); Serial2.print(";");
        }
      else
        {
        Serial2.print(z); Serial2.print(";");
        }
    }
  else
   //k = 1000 to 1440
   //****************
    { 
     z = int(k/100);
     Serial2.print(z); Serial2.print(";");
     
     z = (k - (z*100));  
     if (z <=9)
        {
        Serial2.print("0"); Serial2.print(z); Serial2.print(";");
        }
      else
        {
        Serial2.print(z); Serial2.print(";");
        }
    }
    
  //send 1440 point windspeed data array
  //*************************************
  for (x=0;x<=1440;x++)
    {
    if (todayArray[x] == 0)
        {
         Serial2.print(".3");
        }   
    else if ((todayArray[x] >= 1) and (todayArray[x] <=9))
        {
        Serial2.print("0");
        Serial2.print(todayArray[x]);
        }
    else if (todayArray[x] >= 10)
        {
        Serial2.print(todayArray[x]);
        }
    Serial2.print(";");          //add ";" delimiter after each two digit number
    }


 
 //Send Configuration File
 //************************

 //label 1 - Max Windspeed Title   (not shown on screen at present, see code below)
 //******************************
  //MaxWindspeedTime is called when the maximum is updated in another routine
  //It returns time2
  String InterimMax;
  InterimMax = "set label 1 'TODAY MAX = ";
  if (maxToday <= 9)
    {
    InterimMax.concat("0");
    }
  InterimMax.concat(maxToday);                  //NOTE: THE DASHED LINE GETS ITS VALUE FROM HERE
  InterimMax.concat(" mph (");                  //SO IF THIS IS CHANGED, THE POSITION OF THE MAX
  InterimMax.concat(time2);                     //CHARACTERS MUST BE CHANGED IN THE RASPI.
  //InterimMax.concat("' at 15,57 font ',12' tc rgb 'red';");
  InterimMax.concat(")' at 15,77 font ',12' tc rgb 'red';"); 
       //this "test" label is forced off the top of the screen
       //It still needs to sent to the RPi to keep it stable
  Serial2.print(InterimMax);
  
  //label 2 - Past Minute Max Windspeed Title  (not shown on screen at present, see code below)
  //******************************************
  NowWindspeedTime();       //calculates Now time, returns time1
  String InterimNow;
  InterimNow = "set label 2 'PAST MINUTE = ";
  InterimNow.concat(nowPeakWindspeed);
  InterimNow.concat(" mph (");
  InterimNow.concat(time1);
  //InterimNow.concat(")' at 15,53 font ',12' tc rgb 'blue';");
  InterimNow.concat(")' at 15,77 font ',12' tc rgb 'blue';");
       //this "test" label is forced off the top of the screen
       //It still needs to sent to the RPi to keep it stable
  Serial2.print(InterimNow);
  
  //label 3 - Minute Fid Mark and minuteMax value
  //*********************************************
  Yoffset = (minuteMax + 1);
  //Xoffset = (((hour*60) + minute) - 0);
  Xoffset = (nowPlotPoint - 5); 
  String label3;
  label3 = "set label 3 '<";
  if (minuteMax <= 9)
     {
     label3.concat("0");
     }
  label3.concat(minuteMax);
  //label3.concat("now' at ");
  label3.concat("mph' at ");
  label3.concat(Xoffset);
  label3.concat(",");
  label3.concat(Yoffset);
  label3.concat(" font 'arial bold,14' tc rgb 'blue';");
  Serial2.print(label3);
  
  //label 4 - maxToday Value Printed Above Peak
  //********************************************
  //int Yoffset;
  Yoffset = (maxToday + 7);
  Xoffset = (maxTodayOffset - 115);
  String label4;
  label4 = "set label 4 '";
  if (maxToday <= 9)
    {
    label4.concat("0");
    }
  label4.concat(maxToday);
  label4.concat("mph' at ");
  label4.concat(Xoffset);
  label4.concat(",");
  label4.concat(Yoffset);
  label4.concat(" font 'arial bold,20' tc rgb 'red';");  //the serial print must end with a ";" 
  Serial2.print(label4);
  
  //label 5 - "MAX" title printed above TODAY at Peak
  //**************************************************
  Yoffset = (maxToday + 17);
  Xoffset = (maxTodayOffset - 40);
  String label5;
  label5 = "set label 5 'MAX' at ";
  label5.concat(Xoffset);
  label5.concat(",");
  label5.concat(Yoffset);
  label5.concat(" font 'arial bold,12' tc rgb 'red'");
  label5.concat(";");        //the serial print must end with a ";" 
  Serial2.print(label5);
 
 //label 6 - DATE, printed in upper right
 //***************************************
 String label6 = "set label 6 '";
 label6.concat(month);
 label6.concat("/");
 label6.concat(day);
 label6.concat("/");
 label6.concat(year);
 label6.concat("' at 1210,56 font 'arial bold,16' tc rgb 'blue'"); 
 label6.concat(";");
 Serial2.print(label6);
 
 //label 7 - TIME, printed in upper left
 //***************************************
 String label7 = "set label 7 '";
 label7.concat(time1);
 label7.concat("' at 35,56 font 'arial bold,16' tc rgb 'blue'"); 
 label7.concat(";");
 Serial2.print(label7);
 
 //label 8 - "WINDSPEED" MAIN TITLE, printed at top
 //***************************************
 String label8 = "set label 8 'WINDSPEED' at 480,57 font 'arial bold,24' tc rgb 'black'"; 
 label8.concat(";");
 Serial2.print(label8);
 
 //label 9 - "TODAY" title printed below "MAX" at maxToday Peak
  //************************************************************
  Yoffset = (maxToday + 14);   //7
  Xoffset = (maxTodayOffset - 65);   //50
  String label9 = "set label 9 'TODAY' at ";
  label9.concat(Xoffset);
  label9.concat(",");
  label9.concat(Yoffset);
  label9.concat(" font 'arial bold,12' tc rgb 'red'");
  label9.concat(";");        //the serial print must end with a ";" 
  Serial2.print(label9);
 
 //label 10 - time2 printed below "TODAY" title above maxToday peak
  //******************************************************************
  Yoffset = (maxToday + 11);
  Xoffset = (maxTodayOffset - 100);
  String label10 = "set label 10 '";
  label10.concat(time2);
  label10.concat("' at ");
  label10.concat(Xoffset);
  label10.concat(",");
  label10.concat(Yoffset);
  label10.concat(" font 'arial bold,14' tc rgb 'red'");
  label10.concat(";");        //the serial print must end with a ";" 
  Serial2.print(label10);
   
  //label 11 - "V" title Printed above Peak
  //*****************************************
  Yoffset = (maxToday + 2);
  Xoffset = (maxTodayOffset - 15);
  String label11;
  label11 = "set label 11 'V' at ";
  label11.concat(Xoffset);
  label11.concat(",");
  label11.concat(Yoffset);
  label11.concat(" font 'arial bold,14' tc rgb 'red'");
  label11.concat(";");        //the serial print must end with a ";" 
  Serial2.print(label11);
   
  //major config setups
  //********************
  Serial2.print("set yrange[0:60];");
  Serial2.print("set xrange[0:1440];");
  Serial2.print("set xlabel 'TIME' font ',14';");
  Serial2.print("set ylabel 'WINDSPEED (mph)' font ',14';");
  Serial2.print("set border lt 1 lw 2 lc rgb 'black';");
  Serial2.print("set grid ytics lw 1 lt 1 lc rgb 'black';");
  Serial2.print("set grid xtics lw 1 lt 1 lc rgb 'black';");
  Serial2.print("set ytics font ',13';");
  Serial2.print("set xtics ('am' 0,'1' 60,'2' 120,'3' 180,'4' 240,'5' 300,'6' 360,'7' 420,'8' 480,"
                "'9' 540,'10' 600,'11' 660,'N' 720,'1' 780,'2' 840,'3' 900,'4' 960,'5' 1020,"
                "'6' 1080,'7' 1140,'8' 1200,'9' 1260,'10' 1320,'11' 1380,' pm' 1440) font ',13';");
  Serial2.print("plot 'd.dat' u 1:2 with lines lw 2 lc rgb 'blue' notitle, 'dashline.dat' u"
                " 1:2:(0):(1440) w xerror ps 0 lt 0 lw 3 lc rgb 'red' notitle;");
   
  Serial2.print("pause(61);");    //allows plot to remain on screen for 61 seconds, then it  
                                  //disappears and is replaced by the new plot
  
  //send 100 buffer ";" characters  
  for (x=0;x<=100;x++)
    {
    Serial2.print(";");
    }
    
 Serial2.end();      //end sending Data and Config files
 }
 

//FUNCTION: DETERMINE AVERAGE WINDSPEED DURING PAST MINUTE   (not used at this time)
//*********************************************************
 void DetermineNowAvgWindspeed()
  {
   byte total = 0;
   byte b; 
   for (b=1;b<=24;b++)
     { 
     total = total + (nowMinuteArray[b]);
     }
   nowAvgWindspeed = (total/24);
   nowAvgWindspeed = round(nowAvgWindspeed);    //round off the average number, don't truncate it!
  }
 

//FUNCTION: DETERMINE MAX WINDSPEED DURING PAST MINUTE & LOAD VALUE IN TODAY'S ARRAY
//***********************************************************************************
 void UpdateDailyMax()
{
 byte Interim;
 int maxPositionInArray;
 int TimeOfDay = ((hour * 60) + minute);
 minuteMax = 0;    //starting value for comparison
 
 for (x=0;x<=numberOfReadings;x++)      //check all readings during past minute
    {
    Interim = (nowMinuteArray[x]);
    if (Interim >= minuteMax)
       {
       (minuteMax = Interim);
       maxPositionInArray = x;
       }  
    }
    
 nowPeakWindspeed = minuteMax;   
     
 if (minuteMax >= (maxToday + 1))  //check if Today maximum has increased
    {
     maxToday = minuteMax;      
     audioAlertFlag = 1;
     MaxWindspeedTime();        //calculates current MinuteTime, returns time2 
        
    //print maxToday on small screen in case windspeed is over 60 mph  
     lcd.setCursor(9,1);
     lcd.print("TMax:");
     if (maxToday <= 9)
       {
       lcd.print("0");
       }
     lcd.print(maxToday);
    }
  todayArray[TimeOfDay] = minuteMax; 
 }
 
 
//FUNCTION: CALCULATE TIME OF DAY (time1) 
//****************************************
//Time string for display in upper left corner of plot
 void NowWindspeedTime()
 {
  time1 = Convert24HrTimeToAMPM();  //Time 1 = converted time3 
  nowPlotPoint = ((hour*60) + minute);
 }

 
 
//FUNCTION: CALCULATE TIME OF MAXIMUM DAILY READING (time2)
//*********************************************************
//Time string for display on plot, above maxWindspeed
 void MaxWindspeedTime()
 {
  time2 = Convert24HrTimeToAMPM();        //time2 = converted time3
  maxTodayOffset = ((hour*60) + minute);  //used by label4 in plot
 }
  
  
//FUNCTION: CONVERT 24 HR TIME TO AM/PM STRING (for time1 and time2 above)
//************************************************************************
 String Convert24HrTimeToAMPM()
 {
  if (hour <= 11)
    {
     time3 = "";
       if (hour == 0)
         {
         time3.concat("12");
         }
       else
         {
         time3.concat(hour);
         }
     time3.concat(":");
       if (minute <= 9)
         {
         time3.concat("0");
         time3.concat(minute);
         }
       else
         {
         time3.concat(minute);
         }
     time3.concat("am");
    }
  else
    {
     time3 = ""; 
       if ((hour-12) == 0)
         {
         time3.concat("12");
         }
       else
         {
         time3.concat(hour-12);
         }
     time3.concat(":");      
       if (minute <= 9)
         {
         time3.concat("0");
         time3.concat(minute);
         }
         else
         { 
         time3.concat(minute);
         }
     time3.concat("pm");  
    }
    
   return time3; 
  }


 //FUNCTION: CLEAR DATA IN NOW MINUTE ARRAY
 //*****************************************
 void ResetMinuteArrayData()
 {
   for (x=0;x<=29;x++)
     {
      nowMinuteArray[x] = 0;
     }
 } 
 

// FUNCTION: MEASURE WINDSPEED OF DAVIS ANEMOMETER
// ************************************************
 void MeasureWindspeed()
//Count the number of rotations of the Davis anemometer in 2.25 secs, 10 rotations/2.25sec = 10 MPH
//Pin 47 of the Arduino Mega is permanently associated with timer/counter5
{
  //Using Timer 5 because Timer 0 is used by the board and the count pins of Timers 2,3,4 are NC 
  TCNT5 = 0x00;             //count zeroed
  TCCR5B = 0x06;            //0X06 = falling edge, 0x07 = rising edge
  delay(2250);              //Sample period = 2,250 ms or 2.250 sec
  TCCR5B = 0x00;            //stop counter
  Windspeed = TCNT5;        //transfer current count in Timer/Counter5 to Windspeed variable
  TCNT5 = 0x00;             //reset counter5
  delay(10);
  
  if (Windspeed >= 60)     //for plotting safety
    {
    Windspeed = 60;
    }

  //load current windspeed into small array, ~25 windspeed measurements per minute
  nowMinuteArray[oneReadingIndex] = Windspeed;
  
  //Update current minute maximum value, print to LCD  
  if (Windspeed >= oneMinuteMax + 1)
     {
     oneMinuteMax = Windspeed; 
     lcd.setCursor(6,1);
     if (oneMinuteMax <= 9)
       {
       lcd.print("0");
       }
     lcd.print(oneMinuteMax);
     }
    
  oneReadingIndex++;
  numberOfReadings++;
} 


// FUNCTION: READ UTC TIME AND DATE INTO ARRAY AND CONVERT TO LOCAL TIME
// **********************************************************************
 void ReadAndConvertUTCTime()
  {
     
  //Read SX-28 processed GPS data into array
  //*****************************************
  //NOTE: THE SX28 IS SENDING A LEADING NON-PRINTABLE CHARACTER, SO START COUNTING AT GPSARRAY[1].
  Serial1.begin(9600);
  delay(800);          //wait for the data to come into the serial buffer, it comes in after 650ms
  char time;
  char GPSArray[100] = "3";    //the array must be empty
  int x =0; 
  //Read Serial1 Buffer into the array[]
  while (Serial1.available() >= 1)
   {
    GPSArray[x] = Serial1.read();
    x++;
   } 
  Serial1.end();
  
  //Parse GPS serial array & convert to integers
  //*********************************************
   //THE FIRST ELEMENT[0] IN THE ARRAY IS INVALID SO START AT [1].
   
   //                      Time     Sats    Date 
   //Incoming Data Format: 13:45:06 Sats:08 06/25/2015
   //                               111111111122222222  tens 
   //                      123456789012345678901234567  units
   //                  
   month = (((GPSArray[19]-'0')*10) + (GPSArray[20]-'0'));    //convert char to integer
   day = (((GPSArray[22]-'0')*10) + (GPSArray[23]-'0'));      //(-'0' is same as -48dec) 
   year = (((GPSArray[27]-'0')*10) + (GPSArray[28]-'0'));
   hour = (((GPSArray[1]-'0')*10) + (GPSArray[2]-'0'));
   minute = (((GPSArray[4]-'0')*10) + (GPSArray[5]-'0')); 
   second = 0;
   
  //Print number of satellites to 2x16 LCD
  //***************************************
   lcd.setCursor(9,0);         //position#,line#
   int a;
   for (a=10;a<=16;a++)        //read array[10 to 16]
     {
     lcd.print(GPSArray[a]);
     } 
 
  //Convert UTC to local, with daylight savings correction
  //********************************************
     
  // 03/08/2015: start daylight saving
  // 11/01/2015: end 
  // 03/13/2016: start daylight 
  // 11/06/2016: end  
  // 03/12/2017: start daylight 
  // 11/05/2017: end 
    
    azFlag = 0;
     
   //Adjust UTC for local "hour" time, plus adjust for daylight savings 
   //*****************************************************
    if (TZ == -77)     //Most of Arizona is always on STD time
      {
      TZ = -7;
      azFlag = 1;
      }
    
   //Calculate a Standard Time preview for 2015,2016,2017 expressions below
   //************************************************************** 
    daylightSavFlag = 0;  
    //SThour = UTChour + TZ;   //change "UTChour" to "hour" when moved to master program
    SThour = hour + TZ;
    if (SThour <= -1)
       {
       SThour = SThour + 24;
       }
       
    //Handle exception Time Zones
    //****************************
    if ((TZ == -10)||(azFlag == 1))   
       {
        hour = SThour;
        goto EndHourAdj;   //skip over DST, below
       }
       
    //Daylight Saving (DST) starts and ends at 2:00am
    //****************************************
    UTCZerohour = TZ + 24;  //not applicable to worldwide use
    
    if (((year==15) &&\ 
              (((month==3) && (day>=8) && ((SThour>=2) &&\
                 (SThour<=(UTCZerohour-1)))) ||\
              ((month==3) && (day>=9)) ||\
              ((month>=4) && (month<=10)) ||\
              ((month==11) && (day<=0)) ||\
              ((month==11) && (day==1) && ((SThour==0) ||\
                 ((SThour>=UTCZerohour)&&(SThour<=23)))))) ||\
        ((year==16) &&\
              (((month==3) && (day>=13) && ((SThour>=2) &&\
                (SThour<=(UTCZerohour-1)))) ||\
              ((month==3) && (day>=14)) ||\
              ((month>=4) && (month<=10)) ||\
              ((month==11) && (day<=5)) ||\
              ((month==11) && (day==6) && ((SThour==0) ||\
                ((SThour>=UTCZerohour)&&(SThour<=23)))))) ||\ 
        ((year==17) &&\                                              
              (((month==3) && (day>=12) && ((SThour>=2) &&\     
                (SThour<=(UTCZerohour-1)))) ||\
              ((month==3) && (day>=13)) ||\                     
              ((month>=4) && (month<=10)) ||\
              ((month==11) && (day<=4)) ||\                     
              ((month==11) && (day==5) && ((SThour==0) ||\      
                ((SThour>=UTCZerohour)&&(SThour<=23)))))))  
                
     
       //If Daylight Saving conditions are met as detailed above then do Daylight Saving Time 
       //*********************************************************************************** 
       {  
        daylightSavFlag = 1;  
        //hour = UTChour + TZ + 1;   //change "UTChour" to "hour" when moved to master program
        hour = hour + TZ + 1;
        if (hour <= -1)
          {
          hour = hour + 24;
          }
       }  
    else
       //Else do Standard Time 
       //**********************
       {
       hour = SThour;
       } 
      
    EndHourAdj:  
    
    
  //*********************************************************************************** 
 //   Conversion of UTC time to Local am/pm Time Is Complete, with Daylight Saving & Time Zones
  //*********************************************************************************** 
            
  //Using Local time, convert UTC Day and Month to Local Day and Month 
  //***********************************************************************
  // hour = local
  // day = UTC
  // month = UTC
   
  //Standard Time (winter)
  //**********************
  if (daylightSavFlag == 0)  //Standard Time
    {
     if ((day == 1) && ((hour>=(24+TZ)) && (hour<=23)))
       {
        if (month == 1)
           {
           month = 12;            //backed up month
           year = year - 1;       //backed up a year
           }
        else
           {
           month = month - 1;     //backed up month
           }
          
        //the month is backed up now to compensate for UTC reaching midnight before the US
        
        if ((month==1)||(month==3)||(month==5)||(month==7)||(month==8)||(month==10)||(month==12)) 
           {
           day = 31;
           }
           
        if ((month==4)||(month==6)||(month==9)||(month==11))
           {
           day = 30;
           }
           
        if (month == 2)
           {
           if (year == 16)  //2016 is a leap year
              {  
              day = 29;     //2016 is a leap year
              }
           else
              {
              day = 28;
              }      
           }
        }   
     else   //from day >= 2, until the end of the month
        {
        if ((hour>=(24+TZ)) && (hour<=23))
            {
            day = day - 1;
            }
        }
     
    }
  else  //else DaylightSavingsFlag = 1
    {
   
    //Daylight Saving Time(summer)
    //*****************************
      if ((day == 1) && ((hour>=(24+TZ+1)) && (hour<=23))) 
      //If it's the 1st of the month, we need to go backwards into the previous month 
      //because the UTC time gets to midnight before the US does. And UTC advances the
      //date too, which has to be backed out. 
        {
        if (month == 1)
           {
           month = 12;
           }
        else
           {
           month = month - 1;
           }
        
       //Since the months have different number of days, we have to go backwards to the 
       //to the correct day too, at least for the first of the month.
        if ((month==1)||(month==3)||(month==5)||(month==7)||(month==8)||(month==10)||(month==12)) 
           {
           day = 31;
           }
           
        if ((month==4)||(month==6)||(month==9)||(month==11))
           {
           day = 30;
           }
           
        if (month == 2)
           {
           if (year == 16)
              {  
              day = 29;     //2016 is a leap year
              }
           else
              {
              day = 28;
              }      
           }
        }   
      else   //if the date is NOT the 1st of the month, we do a simple subtract
        {
          if ((hour>=(24+TZ+1)) && (hour<=23))   
            {
            day = day - 1;
            }
        }
     } 
  } 
    
  
//End of program
//***********************************************************************************
 
  

