
;**********************************************************************
;**********************************************************************
;                                    PARALLAX SX28 MICROCONTROLLER
;                        GPS DECODER FOR ANEMOMETER TRACKER PROJECT
;**********************************************************************
;**********************************************************************

;File:  SX28WindspeedGPS_SRC_Word       
;Date: 9/25/2015

;IDE: Parallax SX-Key Editor v3.3.0, available on the Parallax website. 

[bookmark: _GoBack];This program was written in 2015 by David Goodsell.

;This assembly program was written for the Parallax SX28 microcontroller
;and decodes the GMT Time(hrs/min/sec),Date(month/day/year) and number of
;satellites, by parsing the GGA and RMC sentences from a Garmin GPS
;receiver.

;The serial output of Time, Date and satellites is at 5 volt levels, at 
;9600 baud, starting high. The program also supplies a 1 second negative
;going 5 volt pulse at the start of each minute.

;In operation, the GPS LED will flash at 1 PPS when the satellites are
;acquired and also at 1 PPM when plotting windspeed.

;The GPS receiver is a Garmin GPS18x LVC, set for 9600 baud. It has been
;configured to output ONLY the RMC and GGA sentences, at 1 sec intervals.
;
;The SX28 uses virtual serial ports to receive data from the Garmin and
;send data to the Arduino at 9600 baud. The code for the Virual Peripheral
;UARTs was written by Chip Gracey of Parallax Inc. and modified by Craig
;Webb.
; 
;*************************************************************************

;Parallax SX28 Microcontroller specs:
;*************************************
;2K EEPROM
;137 bytes RAM
;50MHz clock

;8 Banks of program memory, 256 bytes/bank
;FSR #$00
;FSR #$20
;FSR #$40
;FSR #$60
;FSR #$80
;FSR #$A0
;FSR #$C0
;FSR #$E0


;INTIALIZATION
;*************
DEVICE  SX28L
DEVICE  TURBO, STACKX, OPTIONX      ;STACKX = Stack, Extended
DEVICE	OSCHS3
IRC_CAL IRC_FAST
LIST	Q = 75
FREQ    50_000_000                  ;50 MHz clock
RESET   Initialize


;I/O PINS
;********
MinutePulseToBoxLED	=	ra.0		;Out, starts low 
MinutePulseToArduino 	=	ra.1		;Out, starts high
SentencesFromGarmin	=	ra.2		;In
TimeDateToArduino	=	ra.3		;Out, ser port starts high
OneSecPulseFromGarmin	=	rb.0	;Input to SX28, plus 1 sec
						;trig pulse to Box LED


;VARIABLES
;*********
org  		$08
temp  		ds  	1   	;$08
temp1	   	ds     	1    	;$09
rx_byte		ds	1	;$10
flags	    	ds    	1    	;$11

org		$10
tx_high   	ds	1	;$10
tx_low		ds	1	;$11
tx_count	ds	1	;$12
tx_divide	ds	1	;$13
tx_divisor	ds	1	;$14		
counter1   	ds       	1	;$15
counter2  	ds    	1	;$16
counter3  	ds    	1	;$17
rx_count	ds	1	;$18
rx_divide	ds	1	;$19

org		$30		;Reserved for 16 byte circulating buffer

rx_flag		=	flags.0	


;*************************************************************************
; 				I/O INTERRUPT 
;*************************************************************************
org $0000

Interrupt				;transmit one bit a time
;Transmit byte, one bit at a time
	mov	FSR, #$10							
	test	tx_count		;any bits left to send?
	jz	Receive			;no byte to send, goto Receive
	dec	tx_divide
	jnz	TxDone
	mov   	tx_divide, #32		;#32 = 9600 baud
	stc				;ready stop bit
	rr	tx_high			;goto next bit
	rr	tx_low			
	dec	tx_count		;count - 1

       ;movb 	TimeDateToArduino, /tx_low.6	;inverted serial output
	movb	TimeDateToArduino, tx_low.6	;serial output to Arduino
 
	jmp	TxDone

;Signal from Garmin GPS:
;TTL Tx idles high    
;Start bit = low
;Sent LSB first 
Receive
;Receive bytes, one bit at a time
	movb	C, /SentencesFromGarmin	;GPS Pin
	test	rx_count
	jnz	RxBit
	mov	W, #9
	sc
	mov	rx_count, W
	mov	rx_divide, #48		;#48=9600 baud.  
RxBit
	djnz	rx_divide, RxDone
	mov	rx_divide, #32		;#32=9600 baud  
	dec	rx_count
	sz
	rr	rx_byte
	snz		
	setb	rx_flag			;flag indicates byte received
RxDone
TxDone
	mov	W, #$5D			;interrupt every 163 clocks  
	retiw


;INITIALIZE I/O PORTS
;********************
Initialize
;Set Port States
	mov 	ra, #%1010 		;0=low, 1=high
	mov	rb, #%00000000
	mov	rc, #%00000000			

;Set Port Directions
	mov	W, #$0F
	mov	M, W
	mov	!ra, #%0100		;1=in, 0=out
   	mov	!rb, #%00000001		
	mov	!rc, #%00000000		

;Clear Interrupt variables	
	clrb	rx_flag
	clr	rx_byte
	mov	FSR, #$10
	clr	tx_count
	clr	rx_count
	Call	@DelaySub1sec		;wait for GPS Header(Copyright) 
	Call	@DelaySub1sec		;message to finish

;Clear $30 Memory where GPS data is to be stored
	mov	FSR, #$30
	clr	$30
	clr	$31
	clr	$32
	clr	$33
	clr	$34
	clr	$35
	clr	$36
	clr 	$37
	clr	$38
	clr	$39
	clr	$3A
	clr	$3B
	clr	$3C
	clr 	$3D
	clr 	$3E
	clr	$3F

;*************************************************************************
;			    PROGRAM START
;*************************************************************************

;***********************************
;RECEIVE GPS SENTENCES FROM GARMIN
;***********************************
GPSLoop1
;clear receive variables
	clrb	rx_flag
	clr	rx_byte
	mov	FSR, #$10
	clr	tx_count
	clr	rx_count
	mov	!option, #%10011111    	;bit 6 = rtcc rollover enabled, 
					;        0 = enabled 
;Wait for Garmin One Second Pulse 
;*********************************
	jnb	OneSecPulseFromGarmin, $

;wait for 1st comma, which is start of RMC sentence
FirstCommaLoop
	jnb	rx_flag, $	;wait for rcvd char
	clrb	rx_flag	
	cjne	rx_byte, #$2C, FirstCommaLoop  	;check for ","

;Load Time into Memory $30-$35
	mov	temp, #$30
LoadTimeLoop
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag	
	mov	FSR, temp
	mov	IND, rx_byte
	inc	temp
	cje	temp, #$36, DoneLoadTime
	jmp	LoadTimeLoop
DoneLoadTime

;Let 7 Commas go by until Date in RMC sentence is reached
	mov	temp, #8
EightCommaLoop
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag
	cjne	rx_byte, #$2C, EightCommaLoop
	decsz	temp
	jmp	EightCommaLoop

;Load Date into Memory $37-$3C
	mov	temp, #$37
LoadDateLoop
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag	
	mov	FSR, temp
	mov	IND, rx_byte
	inc	temp
	cje	temp, #$3D, DoneLoadDate
	jmp	LoadDateLoop
DoneLoadDate

;wait for 2nd $, which is start of GGA sentence
DollarSignLoop
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag	
	cjne	rx_byte, #$24, DollarSignLoop	;check for $

;Let 7 Commas go by until Number of Satellites in GGA sentence is reached
	mov	temp, #7
SevenCommaLoop2
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag
	cjne	rx_byte, #$2C, SevenCommaLoop2
	decsz	temp
	jmp	SevenCommaLoop2

;Load "space" in memory $3D
	mov	FSR, #$30
	mov	$3D, #$20

;Load NumSats into Memory $3E, $3F
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag	
	mov	FSR, #$30
	mov	$3E, rx_byte
	
	jnb	rx_flag, $	;wait for rcvd char	
	clrb	rx_flag	
	mov	FSR, #$30
	mov	$3F, rx_byte

DonePulses

		
;*****************************************************************
;	SEND SERIAL DATA(port ra.0) TO ARDUINO AT 1 SEC INTERVALS
;	SEND START PULSE(port ra.1) TO ARDUINO AT 1 MIN INTERVALS
;*****************************************************************

;WAIT FOR START OF MINUTE
;*************************	
;reset ra.0 every time, it will get set below at 00 secs and 1 min					
;reset each pass, because they won't be set until below code is run       
	
	mov	temp, ra
	clrb	temp.0		;reset ra.0 for 1 second LED 
	setb	temp.1		;reset(high) ra.1 to Arduino at 
	mov	ra, temp	;  at 1 min intervals	
			
	mov	FSR, #$30		
	mov	temp, $34	;load Secs, Tens
	cje	temp, #$30, CheckUnits
	jmp	NotZeroSecs
	;Tens are 0
CheckUnits
	mov	FSR, #$30
	mov	temp, $35	;load Secs, Units
	cje	temp, #$30, DoGPSPulse   
	jmp	NotZeroSecs
	;Seconds are 00
DoGPSPulse			;1 sec pulse to LED on Box and Arduino
	mov	temp, ra
	setb	temp.0
	clrb 	temp.1
	mov	ra, temp

Call @DelaySub104Ms

NotZeroSecs

;SEND DATA TO ARDUINO
;********************* 	
	mov	temp1, #$20	;send space
	Call	@SendByte

	mov	FSR, #$30	;send GMT Hours
	mov	temp1, $30
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $31
	Call	@SendByte

	mov	temp1, #$3A   	 ;":"
	Call	@SendByte

	mov	FSR, #$30	;send Minutes, Tens
	mov	temp1, $32
	Call	@SendByte

	mov	FSR, #$30
	mov	temp1, $33
	Call	@SendByte	;send Minutes, Units 

	mov	temp1, #$3A	;send ":"
	Call	@SendByte

	mov	FSR, #$30	;send Seconds
	mov	temp1, $34
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $35
	Call	@SendByte

	mov	temp1, #$20	;send space
	Call	@SendByte
	
;Send "Sats: "
	mov	temp1, #$53	;S
	Call	@SendByte
	mov	temp1, #$61	;a				
	Call	@SendByte
	mov	temp1, #$74	;t
	Call	@SendByte
	mov	temp1, #$73	;s				
	Call	@SendByte
	mov	temp1, #$3A	;:			
	Call	@SendByte
	
	mov	FSR, #$30	;Print number of sats from sentence GGA
	mov	temp1, $3E
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $3F
	Call	@SendByte

	mov	temp1, #$20	;send space
	Call	@SendByte
	mov	temp1, #$20	;send space		
	Call	@SendByte

	mov	FSR, #$30	;send GMT Month
	mov	temp1, $39
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $3A
	Call	@SendByte

	mov	temp1, #$2F  	;"/"
	Call	@SendByte

	mov	FSR, #$30	;send GMT Day
	mov	temp1, $37
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $38
	Call	@SendByte

	mov	temp1, #$2F	;send "/"
	Call	@SendByte
	mov	temp1, #$32	;send "2"
	Call	@SendByte
	mov	temp1, #$30	;send "0"
	Call	@SendByte

	mov	FSR, #$30	;send GMT Year
	mov	temp1, $3B
	Call	@SendByte
	mov	FSR, #$30
	mov	temp1, $3C
	Call	@SendByte

	Call	@DelaySubTwoMs	;SUPER IMPORTANT: GIVE IT TIME TO SEND THE 
				;                 LAST CHARACTER
	jmp	GPSLoop1


;*************************************************************************
;				PAGE 1
;*************************************************************************
org	$0200			


;SUBROUTINE: DELAY SUB, VARIABLE (260ns to 20580ns)(.26 to 20.58us)
;******************************************************************
;Delay Loops = (DesiredDelayTime(ns) - 180ns/80ns)
;Delay = Loops/80ns + 180ns
DelaySub
	decsz	temp1		;temp1 = number of loops(1 to 255)
	jmp	DelaySub
	retp

;SUBROUTINE: 2 millisecond Delay for 9600 baud UART
;*********************************************************
DelaySubTwoMs
	mov	FSR, #$10
	clr	Counter2	;inner loop, 256 loops
   	mov	Counter1, #100	;variable outer loop, 100 loops
Loop1
	decsz	Counter2	;inner loop
	jmp	Loop1
	decsz	Counter1	;variable outer loop
	jmp	Loop1
	retp

;SUBROUTINE: DELAY SUB, MISC
;*******************************
DelaySub104ms			;~104ms 
	mov	FSR, #$10
	mov	Counter3, #20
	jmp	StartDelay
DelaySub1sec			;~1 second
	mov	FSR, #$20
	mov	Counter3, #200
StartDelay
	clr	Counter1
	clr	Counter2
LoopDelay                        
  	decsz  Counter1
  	jmp    LoopDelay
  	decsz  Counter2
  	jmp    LoopDelay
  	decsz  Counter3		;variable outermost loop
  	jmp    LoopDelay
  	retp

;SUBROUTINE:  SEND BYTE (for UART routine in Interrupt routine)
;***************************************************************
;Incoming byte = temp1
;Interrupt must already be turned on in Calling routine
SendByte
	mov	FSR, #$10
	mov	w, temp1	;NEW: USING TEMP1 FOR INCOMING BYTE, NOT W
Wait
	test	tx_count	;wait for not busy
	jnz	Wait
	mov	tx_high, W
	clrb	tx_low.7	;set start bit
	mov	tx_count, #11	;1 start + 8 data + 2 stop bits (note 2 stop bits)
	retp

